Alexander & Sadiku Practice Problem 10.1
Michael Gustafson

```
> restart
```

Handy functions for dealing with phasors

```
> j := I
```

$$j := \mathrm{I} \tag{1}$$

```
> polard := (mag, angd) → polar(mag, angd*Pi / 180)
```

$$polard := (mag, angd) \rightarrow \mathrm{polar}\left(mag, \frac{1}{180}\, angd\, \pi\right) \tag{2}$$

```
> argumentd := (num) → argument(num) * 180 / Pi
```

$$argumentd := num \rightarrow \frac{180\, \mathrm{argument}(num)}{\pi} \tag{3}$$

```
> listphasors := proc(plist) local k
  for k from 1 to nops(plist[ ]) do
  printf("%s = %f < %f deg\n", lhs(plist[ ][k]), evalf(abs(rhs(plist[ ][k]))),
      evalf(argumentd(rhs(plist[ ][k]))))
   end do end proc:
```

Circuit equations

```
> KCLn1 := -Ia + (V1 - 0) / R1 + (V1 - V2)·j·omega·C = 0
```

$$KCLn1 := -Ia + \frac{V1}{R1} + \mathrm{I}\,(V1 - V2)\,\omega\, C = 0 \tag{4}$$

```
> KCLn2 := (V2 - V1)·j·omega·C + (V2 - 0) / (j·omega·L) + (V2 - b·Vx) / R2 = 0
```

$$KCLn2 := \mathrm{I}\,(V2 - V1)\,\omega\, C - \frac{\mathrm{I}\, V2}{\omega\, L} + \frac{V2 - b\, Vx}{R2} = 0 \tag{5}$$

```
> CTRL := Vx = V1
```

$$CTRL := Vx = V1 \tag{6}$$

Solve circuit equations

```
> MySoln := solve({KCLn1, KCLn2, CTRL}, [V1, V2, Vx])
```

$$MySoln := \left[\left[V1 = \frac{\mathrm{I}\, R1\, \left(\mathrm{I}\,\omega^2\, C\, L\, R2 - \mathrm{I}\, R2 + \omega\, L\right)\, Ia}{-\omega^2\, C\, L\, R2 + R2 + \mathrm{I}\, R2\,\omega\, C\, R1 + \mathrm{I}\,\omega\, L - \omega^2\, L\, C\, R1 + \omega^2\, L\, b\, C\, R1},\right.\right. \tag{7}$$

$$V2 = -\frac{Ia\, L\,\omega\, R1\, (\omega\, C\, R2 - \mathrm{I}\, b)}{-\omega^2\, C\, L\, R2 + R2 + \mathrm{I}\, R2\,\omega\, C\, R1 + \mathrm{I}\,\omega\, L - \omega^2\, L\, C\, R1 + \omega^2\, L\, b\, C\, R1}, Vx$$

$$\left.\left. = \frac{\mathrm{I}\, R1\, \left(\mathrm{I}\,\omega^2\, C\, L\, R2 - \mathrm{I}\, R2 + \omega\, L\right)\, Ia}{-\omega^2\, C\, L\, R2 + R2 + \mathrm{I}\, R2\,\omega\, C\, R1 + \mathrm{I}\,\omega\, L - \omega^2\, L\, C\, R1 + \omega^2\, L\, b\, C\, R1}\right]\right]$$

Define lists for known values

```
> Vals := R1 = 2, R2 = 4, L = 2, C = 0.2, omega = 2, Ia = polard(10, 0), b = 3
```

$$Vals := R1 = 2,\ R2 = 4,\ L = 2,\ C = 0.2,\ \omega = 2,\ Ia = \mathrm{polar}(10, 0),\ b = 3 \tag{8}$$

Find numerical solutions

```
> MySoln := subs(Vals, MySoln)
```

$$MySoln := [[V1 = (0.5660377358 + 0.9811320755\,\mathrm{I})\ \mathrm{polar}(10,\ 0),\ V2 = (1.792452830 + 2.773584906\,\mathrm{I})\ \mathrm{polar}(10,\ 0),\ Vx = (0.5660377358 + 0.9811320755\,\mathrm{I})\ \mathrm{polar}(10,\ 0)]] \quad \textbf{(9)}$$

Find phasors

```
> listphasors(MySoln)
V1 = 11.327042 < 60.018361 deg
V2 = 33.023719 < 57.127091 deg
Vx = 11.327042 < 60.018361 deg
```

Conclusion: v1(t) = vx(t) = 11.327 cos(2 t + 60.01 deg), v2(t) = 33.02 cos(2 t + 57.13 deg)